35/replication_package/dofiles/11_InTextResults.do

***********************************************************************************
// Replication Files
**********************************************************************************
/*
HOW DO BELIEFS ABOUT THE GENDER WAGE GAP AFFECT THE DEMAND FOR PUBLIC POLICY?
Sonja Settele
AEJ:pol
*/
**********************************************************************************

***********************************************************************************
**** Replication of in-text results: 
***********************************************************************************

***********************************************************************************
/* Page 11: "Roughly 20 percent of the respondents hold a prior belief below the ACS-
based value of $74, while another 20 percent hold a belief above the CPS-based value
of $94. The median belief is $81." */
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

gen prior_below_74=prior<74
gen prior_above_94=prior>94

tab prior_below_74
tab prior_above_94

sum prior,d

***********************************************************************************
/* Page 11: "When beliefs are incentivized, (...) respondents spend on average 16 seconds
more on their prior estimate." */
***********************************************************************************

reg timeprior prior1 [pweight=pweight],robust

***********************************************************************************
/* Footnote 15: "Only about one third of individuals in the ACS with a 
Bachelor's degree also hold a higher degree, and females' relative wages remain almost
the same when those indiviudals are included ($75 vs. $74)." */
***********************************************************************************

clear all
set more off

use "$path\data\usa_00025.dta", clear

keep if year==2016

keep if age>17&age<66

// Set missing wages to missing if applicable
replace incwage=. if incwage==9999999
replace incwage=. if incwage==999999

gen female=0 if sex!=.
replace female=1 if sex==2

gen GWG_ACS_45_Bachelormore=. 

gen e_sample=.

* uhrswork = weekly hours worked
* educd==101 -> Bachelor's degree
* classwkr=2 -> Employee

/* Calculate women's wages as a share of male wages in the group of 45-year-old employees with 
AT LEAST a Bachelors degree who work 40 hours per week */

reg incwage female if age==45&empstat==1&uhrswork==40&educd>=101&classwkr==2 [pweight=perwt], robust
mat beta = e(b)
sca feml = beta[1,1]
disp feml
replace e_sample=0
replace e_sample=1 if e(sample)==1
mean incwage if female==0&age==45&empstat==1&uhrswork==40&educd>=101&classwkr==2 [pweight=perwt]
matrix mean=e(b)
mat list mean
sca const=mean[1,1]
replace GWG_ACS_45_Bachelormore = 1+(feml/const) if educd>=101&age==45&empstat==1&uhrswork==40&classwkr==2

tab GWG_ACS_45_Bachelormore


// Share of those with a Bachelor's who hold more than a Bachelor's degree: 
tab educd if educd>=101


***********************************************************************************
/* Page 12: "The difference between the 10th and the 90th percentile of the distribution of 
females' actual relative wages across occupation groups amounts to 18 percentage points." */
***********************************************************************************

clear all
set more off

use "$path\data\usa_00025.dta", clear

keep if year==2016

keep if classwkr==2 			//keep only employees

//wage and salary income, set missings to missing
replace incwage=. if incwage==9999999
replace incwage=. if incwage==999999

gen female=0 if sex!=.
replace female=1 if sex==2

// 26 occupation groups based on ACS				//Name of group										BLS category		
gen occ=1 if occ2010>=10&occ2010<=430				//Management, Business, Science and arts			1
replace occ=2 if occ2010>=500&occ2010<=730			//Business Operations Specialists					2
replace occ=3 if occ2010>=800&occ2010<=950			//Financial Specialists								2
replace occ=4 if occ2010>=1000&occ2010<=1240		//Computer and Mathematical							3
replace occ=5 if occ2010>=1300&occ2010<=1540		//Architecture and Engineering						4
replace occ=6 if occ2010>=1550&occ2010<=1560		//Technicians										4
replace occ=7 if occ2010>=1600&occ2010<=1980		//Life, physical and social science					5
replace occ=8 if occ2010>=2000&occ2010<=2060		//Community and Social services						6
replace occ=9 if occ2010>=2100&occ2010<=2150		//Legal												7
replace occ=10 if occ2010>=2200&occ2010<=2550		//Education, Training and library					8
replace occ=11 if occ2010>=2600&occ2010<=2920		//Arts, design, entertainment, sports and media		9
replace occ=12 if occ2010>=3000&occ2010<=3540		//Healthcare practitioners and technical			10
replace occ=13 if occ2010>=3600&occ2010<=3650		//Healthcare support								11
replace occ=14 if occ2010>=3700&occ2010<=3950		//Protective Service								12
replace occ=15 if occ2010>=4000&occ2010<=4150		//Food preparation and serving						13
replace occ=16 if occ2010>=4200&occ2010<=4250		//Building and grounds cleaning and maintentance	14
replace occ=17 if occ2010>=4300&occ2010<=4650		//Personal care and service							15
replace occ=18 if occ2010>=4700&occ2010<=4965		//Sales and related									16
replace occ=19 if occ2010>=5000&occ2010<=5940		//Office and administrative support					17
replace occ=20 if occ2010>=6005&occ2010<=6130		//Farming, fishing and forestry						18
replace occ=21 if occ2010>=6200&occ2010<=6765		//Construction										19
replace occ=22 if occ2010>=6800&occ2010<=6940		//Extraction										19
replace occ=23 if occ2010>=7000&occ2010<=7630		//Installation, maintenance and repair				20
replace occ=24 if occ2010>=7799&occ2010<=8965		//Production										21
replace occ=25 if occ2010>=9000&occ2010<=9750		//Transportation and material moving				22
replace occ=26 if occ2010>=9800&occ2010<=9830		//Military specific


keep if age>24&age<66

// Drop following industries: 0=N/A, 992=last worked 1984 or earlier, 999=did not respond
drop if ind1990==0|ind1990==992|ind1990==999

//Summarize some occupation groups -> map to Bureau of Labor Statistics categories: 
replace occ=21 if occ==22  			//Extraction: Add to Construction
replace occ=2 if occ==3				//Financial Specialists: Add to business specialists
replace occ=5 if occ==6				//Technicians: Add to architecture and engineering
replace occ=. if occ==26			//Drop military

egen occnew= group (occ)

rename occ occold
rename occnew occ


//BLS categories
*occ=1 			//Management, Business, Science and arts			
*occ=2 			//Business Operations Specialists and Financial Specialists					
*occ=3 			//Computer and Mathematical							
*occ=4 			//Architecture and Engineering and Technicians					
*occ=5 			//Life, physical and social science					
*occ=6 			//Community and Social services					
*occ=7 			//Legal												
*occ=8 			//Education, Training and library					
*occ=9 			//Arts, design, entertainment, sports and media		
*occ=10 		//Healthcare practitioners and technical			
*occ=11 		//Healthcare support								
*occ=12 		//Protective Service								
*occ=13 		//Food preparation and serving						
*occ=14 		//Building and grounds cleaning and maintentance	
*occ=15 		//Personal care and service							
*occ=16 		//Sales and related									
*occ=17 		//Office and administrative support					
*occ=18 		//Farming, fishing and forestry						
*occ=19 		//Construction and Extraction									
*occ=20 		//Installation, maintenance and repair				
*occ=21 		//Production										
*occ=22 		//Transportation and material moving	


// Generate occupation-level measure of females' relative wages
gen GWG_ACS_2565_occ=.
gen e_sample=.

forvalues i= 1/22 {
disp `i'
reg incwage female if empstat==1&uhrswork==40&classwkr==2&occ==`i' [pweight=perwt], robust
mat beta = e(b)
sca feml = beta[1,1]
disp feml
replace e_sample=0
replace e_sample=1 if e(sample)==1
mean incwage if female==0&empstat==1&uhrswork==40&classwkr==2&occ==`i' [pweight=perwt]
matrix mean=e(b)
mat list mean
sca const=mean[1,1]
replace GWG_ACS_2565_occ = 1+(feml/const) if empstat==1&uhrswork==40&classwkr==2&occ==`i'
replace e_sample=.
}

_pctile GWG_ACS_2565_occ [pw=perwt], p(10)
return list
*-> .704116702079773 		

_pctile GWG_ACS_2565_occ [pw=perwt], p(90)
return list
*-> .8796714544296265 

// Diff: 0.1755

***********************************************************************************
/* Page 12: "The difference between the 10th and the 90th percentile of females' actual 
relative wages across states amounts to 8 percentage points." */
***********************************************************************************

gen state=""
replace state="Alabama" if statefip==1
replace state="Alaska" if statefip==2
replace state="Arizona" if statefip==4
replace state="Arkansas" if statefip==5
replace state="California" if statefip==6
replace state="Colorado" if statefip==8
replace state="Connecticut" if statefip==9
replace state="Delaware" if statefip==10
replace state="District of Columbia" if statefip==11
replace state="Florida" if statefip==12
replace state="Georgia" if statefip==13
replace state="Hawaii" if statefip==15
replace state="Idaho" if statefip==16
replace state="Illinois" if statefip==17
replace state="Indiana" if statefip==18
replace state="Iowa" if statefip==19
replace state="Kansas" if statefip==20
replace state="Kentucky" if statefip==21
replace state="Louisiana" if statefip==22
replace state="Maine" if statefip==23
replace state="Maryland" if statefip==24
replace state="Massachusetts" if statefip==25
replace state="Michigan" if statefip==26
replace state="Minnesota" if statefip==27
replace state="Mississippi" if statefip==28
replace state="Missouri" if statefip==29
replace state="Montana" if statefip==30
replace state="Nebraska" if statefip==31
replace state="Nevada" if statefip==32
replace state="New Hampshire" if statefip==33
replace state="New Jersey" if statefip==34
replace state="New Mexico" if statefip==35
replace state="New York" if statefip==36
replace state="North Carolina" if statefip==37
replace state="North Dakota" if statefip==38
replace state="Ohio" if statefip==39
replace state="Oklahoma" if statefip==40
replace state="Oregon" if statefip==41
replace state="Pennsylvania" if statefip==42
replace state="Rhode Island" if statefip==44
replace state="South Carolina" if statefip==45
replace state="South Dakota" if statefip==46
replace state="Tennessee" if statefip==47
replace state="Texas" if statefip==48
replace state="Utah" if statefip==49
replace state="Vermont" if statefip==50
replace state="Virginia" if statefip==51
replace state="Washington" if statefip==53
replace state="West Virginia" if statefip==54
replace state="Wisconsin" if statefip==55
replace state="Wyoming" if statefip==56		


// Generate state-level measure of females' relative wages
replace e_sample=.
gen GWG_ACS_2565_state=.

egen group= group (statefip)
sum group, meanonly

forvalues i= 1/`r(max)' {
disp `i'
reg incwage female if empstat==1&uhrswork==40&classwkr==2&group==`i' [pweight=perwt], robust
mat beta = e(b)
sca feml = beta[1,1]
disp feml
replace e_sample=0
replace e_sample=1 if e(sample)==1
mean incwage if female==0&empstat==1&uhrswork==40&classwkr==2&group==`i' [pweight=perwt]
matrix mean=e(b)
mat list mean
sca const=mean[1,1]
replace GWG_ACS_2565_state = 1+(feml/const) if empstat==1&uhrswork==40&classwkr==2&group==`i'
replace e_sample=.
}

_pctile GWG_ACS_2565_state [pw=perwt], p(10)
return list

_pctile GWG_ACS_2565_state [pw=perwt], p(90)
return list

// Diff: 0.07


***********************************************************************************
/* Footnote 30: "Young individuals are an exception: The treatment effect 
on specific policy demand in the group of 18 to 24-year-olds corresponds to a 
substantial 0.25 standard deviations for female and to zero for male respondents 
(p-value of the difference <0.05). */
***********************************************************************************

clear all

use "$path\data\SurveyStageI_AB_final.dta", clear

global controls wave democrat indep otherpol prior midwest south west anychildren loghhinc associatemore fulltime parttime selfemp unemp student

// Drop control group, keep only individuals aged 18-24
drop if rand==0
keep if age==1

* Split samples by gender: Test for equality of treatment coefficient 
eststo: reg z_lmpolicy_index T1 $controls [aweight=pweight] if female==0
eststo: reg z_lmpolicy_index T1 $controls [aweight=pweight] if female==1
    
suest est1 est2, vce(robust)
test [est1_mean]T1 = [est2_mean]T1


***********************************************************************************
/* Footnote 31: "15, 14 and 36 percent of respondents in the pure control group agree 
that men are inherently more i) ambitious, ii) talented for demanding tasks and iii) interested
 in ``technical'' jobs, respectively. Conversely, 71, 70 and 62 percent believe that i) women 
 face discrimination in the labor market, ii) men have been encouraged more to pursue ambitious 
 careers and iii) women face larger difficulties in combining work and family in today's society." */
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

* keep only pure control group
keep if rand==0

// Answer scale: 1=completely disagree, 2=disagree, 3=neither agree nor disagree, 4=agree, 5=completely agree

gen ambitiousagree=ambitiousraw>3 if ambitiousraw!=.
gen talentedagree=talentedraw>3 if talentedraw!=.
gen interestedagree=interestedraw>3 if interestedraw!=.

tab ambitiousagree
tab talentedagree
tab interestedagree

gen discriminationagree=discriminationraw>3 if discriminationraw!=.
gen encouragedagree=boysraw>3 if boysraw!=.
gen difficultiesagree=societyraw>3 if societyraw!=.

tab discriminationagree
tab encouragedagree
tab difficultiesagree

***********************************************************************************
/* Footnote 34: "34, 30 and 36 percent of the respondents believe that 
anti-discrimination, affirmative action and family policies, respectively, are 
`` somewhat effective'' or ``highly effective'' rather than ``strongly/somewhat 
counterproductive'' or ``neither effective nor counterproductive''." */
***********************************************************************************

clear all
set more off

use "$path\data\SurveyStageI_AB_final.dta", clear

/* Answer scale: 1=strongly counterproductive, 2=somewhat counterproductive, 3=Neither 
effective nor counterproductive, 4=somewhat effective, 5=highly effective */

gen effdisagree=effdisraw>3 if effdisraw!=.
gen effAAagree=effAAraw>3 if effAAraw!=.
gen effworkfamagree=effworkfamraw>3 if effworkfamraw!=. 

tab effdisagree
tab effAAagree
tab effworkfamagree